Make Mine MAPPER #4PRIVATE 

by Rob Haeuser

How Do You Choose Which Function To Use?

________________________________________________________________________

     Sometimes the hardest part of writing runs is deciding exactly which function is most appropriate for the specific need.  Should I use SRH or RDL/IF logic?  For finding a record, we have BFN, FDR, and FND.  There are at least five functions for doing math: ART, CAL, CHG, SUB, and TOT.  What about the new math function, Count?  The number of possibilities available for almost every type of processing boggles the mind.

     As I've said before, there's almost always more than one way to write it.  The driving force behind the decision is ultimately efficiency.  Which technique does it fastest with the least amount of IO?  You might be surprised to find that sometimes the differences are more than just dramatic, they can be downright unbelievable!  So let's look at some coding techniques and why one  way may be far superior to another.

     One of the simplest examples that comes to mind is searching a rid.  You would think that it would be simple enough.  Use SRH, right? Maybe not.  The first question is: Can you get all the data you want in a single SRH?  If so, and no further processing is required, use it.  But, if you have to search a rid and then search the subsequent result (maybe over and over), there could be a better way.  This is often the case if a rid has imbedded blanks that need to be extracted before doing the 'real' search. Figure 1 demonstrates this technique.

________________________________________________________________________

Figure 1

@SRH,0,A,4,6,,7 DN@ 2-1 ],@ SRH,-0,6,,8 D@ 15-5 ],@ .

@SRH,-0,6,,9 DN 34-4 ],OPEN TOT,-0 O 45-3 ],+ DSP,-0 .

@REL .

@7:. error message indicating all records have blank keys

@8:. error message indicating no blanks in 15-5

@9:. error message indicating no 'OPEN' records

________________________________________________________________________

     We want to eliminate all the records that are completely blank and then process any records with blanks in columns 15 through 19 that do not contain the value 'OPEN' in columns 34 through 37.  Finally, we want to add up the values in columns 45 through 47.  The ']' represents the tab character.  The first search eliminates all totally blank records by searching for non-blank keys.  Any record without a key is discarded. The key is actually greater than one character, but never contains blanks, so why search more columns than you absolutely have to?  The second SRH gets all the blanks in 15-5, the third SRH eliminates the 'OPEN' records, and finally we TOT that result.  Note that the TOT uses the 'O' option (we simply want a count without seeing all the data).

     Since in this example most records were ultimately included in the TOT, essentially we processed the data four times.  We could probably have coded a range search to reduce the number of passes, but would still have processed the data more than once. There may be a more efficient way (see Figure 2).

________________________________________________________________________

Figure 2

@BRK LDV V5I5 RDL,0,A,4,6,7 2-1,15-5,34-4,45-3 V1H,V2H,V3H,V4I .

@IF V1 = '' . ;IF V2 NE '' . ;IF V3 = OPEN . ;INC,V4 V5 .

@RLN,,LIN1 2-1,15-5,34-4,45-3 V1,V2,V3,V4 GTO LIN-1 .

                       GRAND TOTAL = V5

@BRK DSP,-0 .

@REL .

@7:. error message indicating no records to process

________________________________________________________________________

     First we BRK to clear the output area, then we initialize our accumulator variable (V5) and read the first record.  The three IF statements assure that only the desired records are processed.  Then the RLN processes all subsequent records until the end of the rid is reached.  Finally, we place the grand total in the output area, break, and display it (the output displayed has been simplified for this example).  The data has been processed once, not four times.  Of course, the LLP for Figure 2 is greater than that for Figure 1, but the I/O is much less. 

     But we're not finished yet.  The code in Figure 2 may be more efficient from an I/O point of view, but it is LLP intensive, and it just plain looks cumbersome.  There is a third alternative.  I vote for the code in Figure 3.

________________________________________________________________________

Figure 3

@CAL,0,A,4,6,,9 LOV 15-5,34-4,45-3 ],A,B,C\

 IF:A='';IF:B<>'OPEN';THEN:TOTAL=TOTAL+C .

@9:. error message indicating no data to process

________________________________________________________________________

     Note that we do not process column 2 (the key) for non-blank data.  This is handled by the 'V' option, which instructs CAL to process only those lines with valid numerical data.  By the way, the code in Figure 2 should probably have done a range check before incrementing V5.  Otherwise, you run the risk of run failure (there's a 'gotcha!').

     The CAL function could almost qualify as a programming language in itself.  I have seen it used in some rather bizarre ways: for instance as a substitute for FDR/RLN, although that was probably a proof of concept effort and amounts to over-kill.  Even if no totaling had been required for our example, CAL is still very effective as a substitute for multiple SRH's or read/if loops.

     One more suggestion for code that can substitute for SRH: if the field(s) being searched are sorted, use binary-find with the 'O' option, which creates a result.  It is definitely faster than searching. 

     A technique I have heard of over the years has always bothered me slightly.  It is where a 'next available line to write' number is maintained in the header of a rid.  This is used as a substitute for find-a-blank (FND or FDR with the '@' option).  What if there are imbedded blanks?  What if somebody manually (heaven forbid!) changes the rid size?  And you have to compare that number somehow to the size of the rid to avoid attempting to write beyond the end of said rid.  Not to mention the fact that every time you write a record you also have to write the header record again, which doubles your write IO.  I can just see that recovery tape spinning madly to keep up with double the number of required writes! Seems like a lot of extra effort to me!  So what about FND and FDR?

     The normal course of events should allow you to write new data to the bottom of the rid.  If this is the case, look at FND.  How many blank lines do you add when you need to?  Generally, I add 100 lines, depending on the application, of course.  When the run is started, check for the last line number using LLN and decrement that number by the amount you would ordinarily add (100 in this case).  Then start the find from that point.  You never scan more than 100 lines, regardless of the size of the rid (see Figure 4).

________________________________________________________________________

Figure 4

@LLN,600,B,55 V8I4 DEC,100 V8 .    get last line number & decrement 

@66:.                              edits or other code would go here

@LOK,600,B,55 FND,600,B,55,100,88 @ 2-1 ],@ ,V8 .

@77:WRL,600,B,55,V8,Y 12-8 ],'Hi there' ULK GTO 66 . 

@88:LLN,600,B,55 V8 LN+,600,B,55,V8,100 INC V8 GTO 77 .
________________________________________________________________________

     If you are required to write data to the top of the rid, FDR is the function of choice because, unlike FND, FDR can control how many lines are scanned.  You can use a variable to control how many lines are scanned, so that every time a hit is made, the variable is decremented to reduce the number of lines processed by the next FDR. Figure 5 illustrates this technique.

________________________________________________________________________

Figure 5

@LDV V8I3=100,V9I4=6 . 

@LOK,126,H,4 FDR,126,H,4,V9,V8,30 @ 2-1 ],@ ,V9 CHG V8 V8 -V9 +5 .

@15:WRL . . . ULK ETC.

@30:LN+,126,H,4,5,100 LDV V8=100,V9=6 GTO 15 .

________________________________________________________________________

     Notice that we decrement V8 by the line number of the hit and then add 5.  This assumes that there are 5 header lines.   I used CHG instead of a DEC/INC combination because it can do what I want in one function instead of two, and (I know this sounds trivial, but) it takes less keystrokes.  In this example, it meant the difference between fitting all the code on one line instead of two.

     Actually, you could just hard-code 100 to avoid the need to maintain V8.  However, IO will be reduced, especially as you use up more and more blank lines. Speaking of blanks, the rid in these examples does not contain blank lines scattered throughout.  If it had, you would probably want to hard-code the starting line number at 6 and just scan the whole rid each time.  I still maintain that read (or find) IO is cheaper than write IO (no tape to update) and therefore more efficient than a double write.

     If you have programmed in another language, such as COBOL, you may suffer from what I call the 'transaction syndrome'.  Even after seven and a half years, I still catch it occasionally.  After all, you 'read' and 'write' records, don't you?  Not if you can help it.

     Just recently I wrote a database initialize run that pulled data from three different sources to construct over 4,000 rids with well over 1,000,000 lines of data.  The main input files are read record-by-record, with two records being converted to three.  As each pair of records was read, I used BFN to gather data from two other rids, one roughly 45,000 lines and the other roughly 3,500.  Seems straightforward enough.  It took longer than I had expected to execute.  Since I can never leave code alone anyway, I re-wrote it with a different approach.

     Using BFN meant that those rids were sorted, and since the rids being constructed were ultimately sorted anyway, it brought up the possibility that I could construct it, then match to each of the two large input rids to move the required data over.  Even though it required two sorts (each of the two input rids had different keys for the match), the I/O used, and therefore execution speed, were dramatically reduced.  I/O was reduced by over 80% and execution time was cut by more than 50%! 

     A good rule of thumb is: never be satisfied.  No matter how good you think the code is, there's probably a more efficient way of doing it.  I will write a run, take I/O and LLP counts, time the execution, and then completely re-write the code for a comparison, sometimes four times or more.  This may sound like a waste of time, but if that run executes more than once a year, it is probably more than worth the effort. Besides, it is a great way to learn the best technique for the task.

     Well, by now you’ve probably guessed that there is no secret method for deciding on a function.  You know what I do?  Every time a new level of MAPPER comes out, I scan the Run Design manual's table of contents.  What new functions did we get?  Kind of like a kid at Christmas.  It may be impossible to memorize nearly 150 different functions and all their syntax, but try to familiarize yourself with them enough so that you can look one up when you need it.  In a word: experiment!  Try different ways. Take IO, LLP, and timings.  The time you save may be the machines!

________________________________________________________________________

